HW #4 Solutions

Here is a trick for defining positive constants. Whenever Simplify or Integrate are used, these assumptions will be applied.

In[33]:=
```
$Assumptions = {ℏ > 0, m > 0, ω > 0, l > 0}
```

Out[33]= $\{\hbar > 0, m > 0, \omega > 0, l > 0\}$

Since we will be calculating ΔxΔp several times over, it is convenient to define a general function for doing so. The Module command allows a function to use a number of different intermediate steps before returning its final value. In this case, the values of $\langle x^2\rangle$, $\langle x\rangle$, and $\langle p^2\rangle$ are first calculated and put in the local variable xsq, xx, and psq. Then ΔxΔp is calculated from those.

In[346]:=
```
ΔxΔp[psi_] := Module[{xsq, xx, psq}, {xsq, xx, psq} =
    Integrate[psi {x^2 psi, x psi, -ℏ^2 D[psi, {x, 2}]}, {x, -∞, ∞}] / Integrate[psi^2, {x, -∞, ∞}]; Sqrt[(xsq - xx^2) psq]
  ]
```

define the simple harmonic oscillator wavefunctions

In[6]:=
```
sho[n_] := With[{y = x / Sqrt[ℏ/(m ω)]}, HermiteH[n, y] Exp[-y^2 / 2]]
```

Problem 1: uncertainty product for first 4 states of the harmonic oscillator

In[46]:=
```
Table[ΔxΔp[sho[n]], {n, 0, 3}] // Simplify
```

Out[46]= $\left\{\frac{\hbar}{2}, \frac{3\hbar}{2}, \frac{5\hbar}{2}, \frac{7\hbar}{2}\right\}$

note that each successive state adds 1 $\hbar$ to the uncertainty product

Problem 2: repeat for the particle in a box.

define the wavefunctions for the PIAB. This uses the Piecewise function to define the wavefunctions everywhere.

In[348]:=
```
piab[n_] := Piecewise[{{0, x < 0}, {Sin[n π x / l], 0 ≤ x < l}, {0, x ≥ l}}]
```

$$\text{piab}[n\_] := \begin{cases} 0 & x < 0 \\ \sin[n\pi x/l] & 0 \le x < l \\ 0 & x \ge l \end{cases}$$

In[349]:=
```
Table[ΔxΔp[piab[n]], {n, 1, 4}] // Simplify
```

Out[349]= $\left\{\frac{1}{2}\sqrt{\frac{1}{3}\left(-6+\pi^2\right)}\,\hbar, \sqrt{\frac{1}{6}\left(-3+2\pi^2\right)}\,\hbar, \frac{1}{2}\sqrt{-2+3\pi^2}\,\hbar, \sqrt{\frac{1}{6}\left(-3+8\pi^2\right)}\,\hbar\right\}$

In[350]:=
```
% // N (*numerical values*)
```

Out[350]= $\{0.567862\,\hbar, 1.67029\,\hbar, 2.6272\,\hbar, 3.55802\,\hbar\}$

notice that the values are pretty close to those of the harmonic oscillator. Each successive energy level adds a little less than 1 $\hbar$.

Problem 3

note that the odd sho wavefunctions are solutions to the Schrodinger Eqn with $\psi(0)$=0, which is exactly what we need for this problem. Thus E=(2n+1+1/2)$\hbar\omega$=(2n+3/2)$\hbar\omega$. The corresponding eigenfunctions are $H_{2n+1}\left(\sqrt{\frac{m\,\omega}{\hbar}}\;x\right)e^{-\frac{m\,\omega x^2}{2\hbar}}$.

Problem 4

now define the half-harmonic oscillator wavefunctions

In[354]:=
```
sho2[n_] :=
  With[{y = x/Sqrt[ħ/(m ω)]}, Piecewise[{{HermiteH[2 n + 1, y] Exp[-y^2/2], y > 0}, {0, y ≤ 0}}]]
```

In[359]:=
```
Table[ΔxΔp[sho2[n]], {n, 0, 3}] // N // Simplify
```

Out[359]= $\{0.583216\,\hbar,\ 1.49105\,\hbar,\ 2.37291\,\hbar,\ 3.24886\,\hbar\}$

Problem 5

In[362]:=
```
Integrate[x sho2[9]^2, {x, 0, ∞}] / Integrate[sho2[9]^2, {x, 0, ∞}] // Simplify // N
```

Out[362]= $\frac{3.97634\,\hbar}{\sqrt{m\,\omega\,\hbar}}$

classical solution: $x(t) = x_T|\sin\omega t|$. Classical turning pt at energy E is $x_T=\sqrt{\frac{2E}{m\omega^2}} = \sqrt{\frac{2\hbar\omega(2n+3/2)}{m\omega^2}} = \sqrt{\frac{39\,\hbar}{m\,\omega}}$.

The time average is

In[365]:=
```
Integrate[Sqrt[39 ħ/(m ω)] Sin[ω t]/(π/ω), {t, 0, π/ω}] // N
```

Out[365]= $3.97569\sqrt{\frac{\hbar}{m\,\omega}}$

which is amazingly close. Both classical and quantum theories of harmonic oscillators give similar answers, except for the key difference which is only certain energies being allowed.